
ABSTRACT  

The use of online platforms such as Survey Monkey and Qualtrics allow for ease of survey administration. 
Survey items are entered into the system and the data are collected directly from respondents 
electronically. While these online platforms provide summaries of results, the reports are often not in a 
user-friendly format that is ready to share with stakeholders. The purpose of our survey report macro is to 
prepare descriptive summaries of item-level results from surveys collected from online systems in a user-
friendly format. Features of the macro code include data steps using arrays and do loops to organize the 
data for analysis, PROC TABULATE to produce accessible tables of results, and use of the Output 
Delivery System (ODS) to create reports in Word or Excel format. 

The survey report macro produces survey results for Likert-type items from a single survey administration. 
In our work as program evaluators, our projects typically include multiple schools that are implementing a 
common program. We often wish to summarize survey results for all schools within the project to inform 
the program overall and for individual schools to share with school personnel for use in program planning 
at the school-level. The survey report macro allows flexibility to run analyses for different subsets of 
participants for whom the user wishes to report results. There are also user-specified choices for reporting 
options built into the macro code. 

INTRODUCTION  

We developed the code described in this paper in order to summarize descriptive survey results for 
various subsets of the survey data. For a project that include multiple schools, we often need to prepare 
summaries at various levels. For example, the project director is interested in overall results across all 
project schools. Summarizing in this manner will provide information on general supports that the project 
director can provide to help all school to improve their survey results. Another stakeholder group is school 
principals. The principals are most concerned about the results at their individual schools and helping 
their school teams to make changes that may improve future survey results. Therefore, we typically run 
reports for the project overall and for each participating school. 

This paper describes how to prepare data for running the code provided in the appendix to produce user-
friendly reports. Our code includes various options, such as collapsing across two or more categories and 
adding item means to the output. In addition, the code offers different sorting options, such as the order 
that items were presented in the original survey or soring in order of frequency of a collapsed category. 
The code may be applied to items that use either a four-point or five-point Likert-type scale. We illustrate 
the process through an example from a four-year project that included implementation of project based 
learning. This project was implemented in 17 schools in four school districts across PK-12 schools. 
Sources for ideas for our code include Delwiche and Slaughter, 2012; Haworth Lake & McKnight, 2015; 
and Gravely, 1998.      

DATA PREPARATION 

To apply the SURVEYREPORT macro, the data need to be set up with survey items in columns and 
participant responses in rows, which is typically how online survey systems provide data. In addition, 
format statements need to be provided for response scales and survey items. In order to apply the macro, 
the variable names need to be labeled x1 to xn where n is the last item on the survey. A codebook should 
be created to connect the x variable names to the descriptive variable names (i.e., the survey items). The 
base macro, called SURVEYREPORT, is called with specified arguments that identify the columns of data 
to be read for a certain subscale or section of the survey. Another macro, called LONGREPORT, is used 
to call the SURVEYREPORT macro multiple times for various subscales on the survey.   

Steps for preparing survey reports with the SURVEYREPORT and LONGREPORT macros: 

1. Prepare the data by downloading from Survey Monkey as an Excel or CSV file. The data will be 
in a wide format with survey items in columns and survey responses in rows. Add x1 to xn 
variable labels to the top row, copy rows with variable labels and survey questions and paste in a 
separate worksheet in Excel using transpose to create the codebook. In the worksheet to be used 
in the macro, remove the rows with survey questions and keep only the x variables as row 
headers/variable names. Delete any open-ended questions from the survey and move to a 
separate file for qualitative analysis elsewhere. See Figure 1 for an excerpt of how the data will 
look when prepared for the SURVEYREPORT macro and Figure 2 for an example of the data 
codebook.  

2. Read the macro-ready data into SAS (x variables and responses). 

3. Set up formats to be created in SAS using appropriate labels for all response scales on the 
survey (e.g., agreement, frequency and survey items). These can be copied and pasted from the 
codebook in Excel. The survey items can be separated into subscales or all listed together. 

4. Select and run the SURVEYREPORT macro to store it in the SAS session. The following steps 
happen inside the macro:  

a. Set up formatting preferences 

b. Apply array statements and do loops to arrange item responses for analysis 

c. Run PROC TABULATE to produce tables of results according to specified options 

5. The LONGREPORT macro is set up to call the SURVEYREPORT macro multiple times and to 
produce output results as RTF/Word or CSV/Excel file using the ODS.      

The SURVEYREPORT macro will run analysis on a set of items within a survey scale or subscale. As 
many of our surveys have multiple item sets/subscales, we commonly need to run the SURVEYREPORT 
macro multiple times. In order to create our survey reports, we use another short macro to call the various 
runs of the SURVEYREPORT macro and output the results using the ODS. The LONGREPORT macro 
arguments include the name of the file, the path to which the file will be saved, and the type of file desired 
(e.g., rtf for a Word document or csv for an Excel file).  

 

x3 x4 x5 x6 x7 x8 x9 x10 x11 x12 x13 x14 x15 x16 

1 1 1 1 1 2 2 1 1 1 1 1 1 2 

1 2 1 1 4 4 4 2 3 3 3 3 1 3 

1 2 1 1 1 4 3 3 4 1 2 2 1 4 

1 1 1 1 1 4 1 1 1 1 1 4 1 1 

1 2 2 2  4 3 2 4 3 4 3 4 2 
 

Figure 1. Excerpt of Data in Wide Variable Format for Five Respondents 

x3 Considering the current school year, 
indicate how often your students do each of 
the following. 

Compare information from various sources 
before completing a task. 

x4 
 

Draw their own conclusions based on analysis 
of numbers, facts, or relevant information. 

x5 
 

Analyze competing arguments, perspectives, or 
solutions to a problem. 

x6 
 

Develop a persuasive argument based on 
supporting evidence or reasoning. 

x7 
 

Assess the accuracy of digital resources. 

x8 
 

Work in small groups to complete an 
assignment together. 

x9 
 

Work with other students to set goals for their 
team. 

x10 
 

Work with other students to create a plan for 
their team. 

x11 
 

Complete collaborative assignments using 
contributions from each team member. 

x12 
 

Provide targeted feedback to peers on their 
academic work. 

x13 
 

Use idea-creation techniques such as 
brainstorming or concept mapping. 

x14 
 

Modify an approach to a problem by testing 
different ideas. 

x15 
 

Invent a solution to a complex, open-ended 
question or problem. 

x16 
 

Create an original product, presentation, or 
performance to express their ideas. 

 

Figure 2. Excerpt of Data Codebook with Response Options from Online Survey System 

 

The macro arguments that must be provided by the user are described below.       

dataset = Name assigned to the dataset read in by the user          

scale = Specify the number of scale points and the number of categories over which to collapse. The 
macro provided specifies options for 42, 52, and 53. The first number specifies the number of scale 
points (e.g., 4 for a 4-point scale) and the second number specifies the number of upper level 
categories in the scale over which to collapse (e.g., 2 for the upper two categories).                                              

mean = If user wants the item mean to appear in the table, indicate this by entering '1'. Otherwise, 
this argument should be set to '0'.  

combine = If the use wants the combined upper categories to appear in the table, indicate this by 
entering ‘1’. Otherwise, this argument should be set to '0'.       

first = Number of the first column in the dataset that holds survey responses that should be included 
in the table                    

last = Number of the last column in the dataset that holds survey responses that should be included in 
the table.                    

Rformat = Name of the user-assigned format (or "value") in proc format for the desired RESPONSE 
SCALE.  

Iformat = Name of the user-assigned format (or "value") in proc format for the desired ITEM.  

category = Name of the user-specified string value to define the collapsed upper categories.  

statement = Item directions or general item statement prefix. This will appear as a header above the 
items in the table.  

order = The desired order by which the items should appear. If the user specifies internal, then the 
items will appear as presented on the survey. If the user specifies data, then the items will appear in 
descending order of frequency by the values of the collapsed upper category.   

SURVEYREPORT AND LONGREPORT MACROS 

The reader may refer to the appendix to follow along with the following description of the macro code. In 
the SURVEYREPORT macro, the data are first transformed from the format as produced from the survey 
software to a long format for analysis. An array is created from a set of items within a survey section. 
Next, a do loop is used to create responses in a long format. Two identical variables of item responses, 
named RESPONSE and RESPONSE2, are created for use in the PROC TABULATE analysis. In 
addition, an UPPER variable is created that represents the highest categories of interest in an analysis. 
For example, we are often interested in the agreement side of a four-point scale that includes strongly 
disagree, disagree, agree, and strongly agree. We add the percentages in the agree and strongly agree 
categories to obtain an overall agreement rate. The UPPER variable is defined as 100 or 0, depending on 
if the observation is in the upper end of the scale. The mean of the UPPER variable is calculated and 
represents the sum of the percentages in the upper categories (idea from pp. 183-185 of Gravely, 1998). 
PROC SUMMARY is used to obtain the mean of the UPPER variable, which is then used as a sorting 
variable named SORTV (idea from pp. 199-201 of Gravely, 1998). Sorting by this variable is useful for 
creating output from PROC TABULATE that is ordered by frequency. The long version of the data set 
called LONGTABLE contains five variables: ITEM, RESPONSE, RESPONSE2, UPPER, and SORTV. 
Data from each respondent is presented in rows associated with each survey item. For each item, there 
are n (the number of respondents) rows. The total number of rows in the LONGTABLE data set is n*m, 
where m is the number of items in the survey section. See Figure 3 for an excerpt of how the data will 
look when transposed into this long format. 

Our sample code includes both four- and five-point scales. For the four-point scales, we specify that the 
upper two categories are added if the user chooses the COMBINE=1 option to produce collapsed results. 
For the five-point scales, we provide choices to either collapse the upper two or three categories when 
collapsed results are desired.  

PROC TABULATE (described in Delwiche & Slaughter, 2012 and Haworth Lake & McKnight, 2015) is 
used to compute and display summary statistics. The variable RESPONSE is specified as a class 
variable for computing frequencies and RESPONSE2 is specified as an analysis variable for computing 
means and/or sample sizes. We use separate order statements for ITEM and RESPONSE variables. The 
order of the items is an argument that can be specified by the user. ORDER=INTERNAL will produce a 
table in the same order as the original data. ORDER=DATA will produce a table in the order of the 
percentage in the UPPER categories, which have been sorted as described above. The RESPONSE 
variable uses internal order to present the categories of the response scale in the order from the survey. 
In addition, PRELOADFMT is used to ensure that all response categories are represented, even if there 
were no respondents who selected one or more of the categories. 

The SURVEYREPORT macro includes options for whether the user wishes to include collapsed 
categories as well as whether the user wishes to include item means. IF THEN statements based on user 
specified arguments are used with do loops to the appropriate PROC TABULATE set up to produce the 
desired output. We find that collapsed categories give a better overview of the survey findings and 
typically include these in our survey reports. 

The LONGREPORT can be set up to make multiple calls to the SURVEYREPORT macro and produce a 
file of results through the Output Delivery System (ODS; described in Delwiche & Slaughter, 2012). 
Options for ODS output are included for both a word processing document using a .RTF extension or a 
spreadsheet format using a .CSV extension. The SURVEYREPORT macro may be called within the 
LONGREPROT macro for various sections of a survey. Using all items on a survey will produce a 
complete report. The results produce reports that are organized in a format that may be shared with lay 
audiences. Table 1 provides an example of results with the top three categories from a five-point scale 

collapsed and no item means requested. Table 2 provides an example of results with the top two 
categories from a four-point scale collapsed with item means included.  

The advantage of creating macro code is that the program may be run multiple times for various ways 
that we wish to examine results. The SURVEYREPORT and LONGREPORT macros can be used in 
combination to produce output for various subsets of the data that we would like to examine. For 
example, consider a project that has 17 schools in four school districts with elementary, middle, and high 
schools. Using various subsets of the data, we can produce results at the project level for the 
management team, at the district level for district administrators, and at the school level for school 
principals. The key demographics must be defined by one or more variables in a data set and then a 
subsetting IF statement can be used to isolate to the subgroup of interest.        

 

ITEM  sortv  RESPONSE upper  response2 

8  88.06723 2 0 2 

8  88.06723 4 100 4 

8  88.06723 4 100 4 

8  88.06723 4 100 4 

8  88.06723 4 100 4 

4  81.98653 1 0 1 

4  81.98653 2 0 2 

4  81.98653 2 0 2 

4  81.98653 1 0 1 

4  81.98653 2 0 2 

13  76.30252 1 0 1 

13  76.30252 3 100 3 

13  76.30252 2 0 2 

13  76.30252 1 0 1 

13  76.30252 4 100 4 

5  75.42088 1 0 1 

5  75.42088 1 0 1 

5  75.42088 1 0 1 

5  75.42088 1 0 1 

5  75.42088 2 0 2 
. 
. 
.   
7  57.31092 1 0 1 

7  57.31092 4 100 4 

7  57.31092 1 0 1 

7  57.31092 1 0 1 

7  57.31092 0   

Figure 3. Excerpt of Data in Long Format for Five Respondents

Considering the current school year indicate how 
often your students do each of the following. 

Never 

A few 
times a 

semester

1-2 times 
per 

month 
1-3 times 
per week Daily 

Monthly 
+ Weekly 
+ Daily  

% % % % % % n 

Work in small groups to complete the assignment 
together 

2.4 9.7 19.6 41.5 26.9 87.9 588 

Draw their own conclusions based on analysis of 
numbers, facts, or relevant information 

4.4 13.7 16.4 39.7 25.8 81.9 592 

Use idea-creation techniques such as brainstorming 
or concept mapping 

8.1 15.6 25.8 36.6 13.9 76.3 590 

Analyze competing arguments, perspectives, or 
solutions to a problem 

7.5 17.3 27.1 32.9 15.3 75.3 590 

Complete collaborative assignments using 
contributions from each team member 

7.6 19.2 27.8 34.4 11.0 73.2 590 

Compare information from various sources before 
completing a task 

7.7 23.2 22.7 36.4 9.9 69.0 594 

Work with other students to set goals for their team 12.2 20.4 28.0 28.0 11.4 67.4 589 

Develop a persuasive argument based on supporting 
evidence or reasoning 

9.7 23.2 30.3 27.9 8.9 67.1 587 

Work with other students to create a plan for their 
team 

13.3 19.8 31.0 26.7 9.2 67.0 587 

Provide targeted feedback to peers on their academic 
work 

11.7 21.4 27.5 29.2 10.2 66.9 589 

Modify an approach to a problem by testing different 
ideas 

12.8 22.3 27.9 27.0 10.0 65.0 588 

Invent a solution to a complex, open-ended question 
or problem 

12.9 23.8 25.2 27.8 10.3 63.3 583 

Create an original product, presentation, or 
performance to express their ideas 

8.7 29.5 31.3 21.6 8.9 61.8 587 

Assess the accuracy of digital resources 18.9 24.0 22.1 22.8 12.1 57.1 587 

Table 1. Example from Output as an RTF File with Data using a Five-Point Scale   

Thinking about this school year to what extent do 
you agree with the following statements? I 
intentionally... 

Strongly 
Disagree Disagree Agree 

Strongly 
Agree 

Agree + 
Strongly 

Agree  

% % % % % Mean n 

include strategies to advance students' critical 
thinking skills in my lesson plans. 

1.2 2.4 56.7 39.8 96.4 3.4 591 

assess students' critical thinking skills. 1.2 3.4 57.3 38.2 95.4 3.3 592 

include strategies to advance students' collaborative 
skills in my lesson plans. 

1.2 5.9 58.5 34.3 92.9 3.3 591 

include strategies to advance students' life skills in my 
lesson plans. 

1.9 6.9 52.7 38.5 91.2 3.3 590 

include strategies to advance students' creativity in 
my lesson plans. 

1.2 9.1 54.7 35.0 89.7 3.2 592 

assess students' collaborative skills. 1.5 9.0 59.8 29.7 89.5 3.2 590 

assess students' creativity. 2.0 13.7 50.2 34.1 84.2 3.2 590 

Table 2. Example from Output as an RTF File with Data using a Four-Point Scale   

CONCLUSION 

This paper demonstrates a process for producing descriptive survey results using a variety of tools 
available in SAS®. Data steps that include the use of arrays and do loops are used to transpose data from 
a wide to a long format in preparation for analysis. PROC SUMMARY is used to create a sorting variable 
to give the user the option to produce results in order of frequency within an item set. The use of PROC 
TABULATE creates appealing tables of results that may include percentages across scale categories and 
may also include item means. The Output Delivery System is used to produce reports in either RTF (for 
word processing software such as Word) or CSV (for spreadsheet software such as Excel) formats. The 
code can be updated to add additional statistics, such as standard deviations, and scales with more than 
five categories.  

A limitation of this process is that the survey items need to be entered into the program for each new 
survey. This can be streamlined by copying and pasting from spreadsheets where the codebooks are 
stored. While the initial set up of items may seem tedious, the payoff is the ability to run analyses for a 
variety of data subsets. This is particularly useful in our work as program evaluators where projects often 
include multiple schools. We can easily prepare results for all schools as well as for each school using 
some basic subsetting data steps.  

APPENDIX 

/************************************************************************** 
/*                    ++++++++++++++++++++++++++++++++++++                   
/*                    ++     SURVEYREPORT MACRO         ++                   
/*                    ++++++++++++++++++++++++++++++++++++                  
/*            
  
/*  This is a macro that uses proc tabulate to generate summary tables for 
reports.  
/* The code used in this macro is a compilation of the work of Drs. 
Tammiee Dickenson, Jessalyn Smith, and Grant Morgan.  
/*  The code in the accompanying LONGREPORT macro was developed by William 
Zachary Smith.  
/*            
  
/*  SURVEYREPORT requires that the user provide each of the following: 
  
/*  dataset = Name assigned to the dataset read in by the user.          
/*  scale = Specify the number of scale points and the number of categories 
over which to collapse.  
 The macro provided specifies options for 42, 52, and 53. The first 
number specifies the number of scale points (e.g., 4 for a 4-point scale)  
 and the second number specifies the number of upper level categories 
in the scale over which to collapse (e.g., 2 for the upper two categories).  
/*  mean = If user wants the item mean to appear in the table, indicate this 
by entering '1'. Otherwise, this argument should be set to '0'.  
/*  combine = If the use wants the combined upper categories to appear in 
the table, indicate this by entering ‘1’. Otherwise, this argument should be 
set to '0'.       
/*  first = Number of the first column in the dataset that holds survey 
responses that should be included in the table.                    
/*  last = Number of the last column in the dataset that holds survey 
responses that should be included in the table.                    
/*  Rformat = Name of the user-assigned format (or "value") in proc format 
for the desired RESPONSE SCALE.  
/*  Iformat = Name of the user-assigned format (or "value") in proc format 
for the desired ITEM.  
/*  category = Name of the user-specified string value to define the 
collapsed upper categories.  
/*  statement = Item directions or general item statement prefix. This will 
appear as a header above the items in the table.  
/*  order = The desired order by which the items should appear. If the user 
specifies internal, then the items will appear as presented on the survey. 
If the user specifies data, then the items will appear in descending order 
of frequency by the values of the collapsed upper category. 
 
/************************************************************************** 
/*  STEP 1:  READ IN THE SURVEY RESPONSES INTO A SAS DATASET                 
*************************************************************************** 
*/ 
 
option nodate nonumber; 
 
*Change the datafile and dbms type if needed (i.e., XLS, CSV) to reflect the 
location of the data on your computer. The header rows should contain x1 to 
xn as the variable names and remaining rows should contain survey 



10 

responses.; 

proc import out=work.beta 

 datafile = "PATH ON COMPUTER\DATAFILE NAME.xls" 
 dbms = xls replace; 
 sheet="MacroFinal"; 
 getnames = YES; 
run; 
 
*Example to subset the data to run results for a particular school – the x2 
variable defines various schools; 
data school1; 
 set beta; 
 if x2=1; 
 run; 
 
/* ************************************************************************* 
/*  STEP 2:  PROVIDE FORMATS FOR RESPONSE SCALES AND ITEMS AND SUBMIT        
/*  NOTE - SAS requires that the names of format values begin and end with 
letters.   
/***************************************************************************
*/ 
 
*You will need to make changes here based on your survey - note that all 
value names and categories can be changed. 
Include all response scales and items on the survey. In this example, the 
survey items are organized by sections that use common scales. For example, 
questiontwo items use the agreement scale and questionthree items use the 
howoften scale. 
; 
proc format; 
 value agreement 
  4="Strongly Agree" 
  3="Agree" 
  2="Disagree" 
  1="Strongly Disagree"; 
 value howoften 
  5="Daily" 
  4="1-3 times per week" 
  3="1-2 times per month" 
  2="A few times a semester" 
  1="Never"; 
  
 value school 
  1="School 1" 
  2="School 2" 
  3="School 3" 
  [MORE SCHOOL NUMBERS OMITTED FROM THIS CODE] 
 
 value questiontwo 

3="Compare information from various sources before completing a 
task" 
4="Draw their own conclusions based on analysis of numbers, 
facts, or relevant information" 
5="Analyze competing arguments, perspectives, or solutions to a 
problem" 



11 

6="Develop a persuasive argument based on supporting evidence or 
reasoning" 

  7="Assess the accuracy of digital resources" 
  8="Work in small groups to complete the assignment together" 
  9="Work with other students to set goals for their team" 
  10="Work with other students to create a plan for their team" 

11="Complete collaborative assignments using contributions from 
each team member" 

  12="Provide targeted feedback to peers on their academic work" 
13="Use idea-creation techniques such as brainstorming or 
concept mapping" 

  14="Modify an approach to a problem by testing different ideas" 
15="Invent a solution to a complex, open-ended question or 
problem" 
16="Create an original product, presentation, or performance to 
express their ideas"; 

value questionthree 
17="include strategies to advance students' critical thinking 
skills in my lesson plans." 

  18="assess students' critical thinking skills." 
19="include strategies to advance students' collaborative skills 
in my lesson plans." 

  20="assess students' collaborative skills." 
21="include strategies to advance students' creativity in my 
lesson plans." 

  22="assess students' creativity." 
23="include strategies to advance students' life skills in my 
lesson plans."; 

[MORE QUESTIONS/ITEMS WERE OMITTED FROM THIS CODE] 
run; 
 
/************************************************************************** 
/*  STEP 3:  READ IN THE SURVEYREPORT MACRO CODE 
*************************************************************************** 
*/ 
 
*Start surveyreport macro; 
%macro 
surveyreport(dataset,scale,mean,combine,first,last,Rformat,Iformat,category,
statement,order); 
 
proc template; 
 define style Styles.OPE; 
 parent = Styles.journal; 
    style Header from HeadersAndFooters /           
           font=('Arial, Arial, Arial',11pt) 
   borderbottomwidth=2.5 
   bordertopwidth=2.5 
   font_weight=bold 
   ; 
    style RowHeader from HeadersAndFooters /        
           font=('Arial, Arial, Arial',11pt) 
   ;     
 style table / 
         borderbottomwidth=2.5; 
end;run; 

data class; 
 do response=1,2,3,4; 
 output; 
 end; 
 run; 
 
%let total = %eval((&last-&first)+1); 
 
data analyze; 
 set &dataset; 
 keep x&first-x&last; 
run; 
 
*Code for combining top 2 of 4 categories; 
%if &scale=42 %then %do; 
 data table; 
  set analyze; 
 run; 
 
data longtable1; 
 set table; 
 array new{&first:&last} x&first-x&last; 
 DO I=&first TO &last; 
 ITEM=I; 
  RESPONSE=new{I}; 
   
  if new{I} in (3,4) then upper=100; 
   else if new{I} in (1,2) then upper=0; 
   
  response2=response; 
  output; 
  end; 
 keep item response response2 upper;  
 run; 
 
proc summary data=longtable1 nway noprint; 
 class item; 
 var upper; 
 output out=dummy(keep=item sortv) mean=sortv; 
 run; 
 
proc sort data=dummy; by item; run; 
proc sort data=longtable1; by item; run; 
 
data longtable; 
 merge dummy longtable1 (in=a);  
 by item; 
 if a; 
run; 
 
proc sort data=longtable; by descending sortv;  
run; 
  
%end; 
 
*Code for combining top 2 of 5 categories; 
%if &scale=52 %then %do; 

 data table; 
  set analyze; 
 run; 
data longtable1; 
 set table; 
 array new{&first:&last} x&first-x&last; 
 DO I=&first TO &last; 
 ITEM=I; 
  RESPONSE=new{I}; 
   
  if new{I} in (4,5) then upper=100; 
   else if new{I} in (1,2,3) then upper=0; 
   
  response2=response; 
  output; 
  end; 
 keep item response response2 upper;  
 run; 
 
proc summary data=longtable1 nway noprint; 
 class item; 
 var upper; 
 output out=dummy(keep=item sortv) mean=sortv; 
 run; 
 
proc sort data=dummy; by item; run; 
proc sort data=longtable1; by item; run; 
 
data longtable; 
 merge dummy longtable1 (in=a);  
 by item; 
 if a; 
run; 
 
proc sort data=longtable; by descending sortv;  
run; 
 
%end; 
 
*Code for combining top 3 of 5 categories; 
%if &scale=53 %then %do; 
 ddata table; 
  set analyze; 
 run; 
data longtable1; 
 set table; 
 array new{&first:&last} x&first-x&last; 
 DO I=&first TO &last; 
 ITEM=I; 
  RESPONSE=new{I}; 
   
  if new{I} in (3,4,5) then upper=100; 
   else if new{I} in (1,2) then upper=0; 
   
  response2=response; 
  output; 
  end; 

 keep item response response2 upper;  
 run; 
 
proc summary data=longtable1 nway noprint; 
 class item; 
 var upper; 
 output out=dummy(keep=item sortv) mean=sortv; 
 run; 
 
proc sort data=dummy; by item; run; 
proc sort data=longtable1; by item; run; 
 
data longtable; 
 merge dummy longtable1 (in=a);  
 by item; 
 if a; 
run; 
 
proc sort data=longtable; by descending sortv;  
run; 
%end; 
 
%if &combine=1 %then %do;  
%if &mean=1 %then %do; 
options orientation=landscape; 
 
proc tabulate data=longtable out=temptable; 
 format item &Iformat.. response &Rformat.. response2 &Rformat..; 
 class response / preloadfmt order=internal; 
 class item / order=&order; 
 classlev item / S=[just=l cellwidth=105 vjust=T]; 
 var response2 upper; 
 classlev response/ style=<parent>; 
 table item=' '*[style=<parent>[just=c]],  response=' '*(pctn 
<response> ='%'*f=5.1)  
 upper=&category*(mean='%'*f=4.1) response2=' '*(mean='Mean'*f=4.1 
n='n'*f=4.)/ 
 RTS=50  printmiss isstext='0.0' BOX={LABEL="&statement" Style={JUST=l 
VJUST=t cellwidth=550}}; 
 title ' '; 
run; 
%end; 
 
 
%if &mean=0 %then %do; 
options orientation=landscape; 
 
proc tabulate data=longtable out=temptable; 
 format item &Iformat.. response &Rformat.. response2 &Rformat..; 
 class response / preloadfmt order=internal; 
 class item / order=&order; 
 classlev item / S=[just=l cellwidth=105 vjust=T]; 
 var response2 upper; 
 classlev response/ style=<parent>; 
 table item=' '*[style=<parent>[just=c]],  response=' '*(pctn 
<response> ='%'*f=5.1)  
 upper=&category*(mean='%'*f=4.1) response2=' '*(n='n'*f=4.)/ 

 RTS=50  printmiss misstext='0.0' BOX={LABEL="&statement" Style={JUST=l 
VJUST=t cellwidth=550}}; 
 title ' '; 
run; 
%end; 
%end; 
 
 
%if &combine=0 %then %do; 
%if &mean=1 %then %do; 
options orientation=landscape; 
 
proc tabulate data=longtable out=temptable; 
 format item &Iformat.. response &Rformat.. response2 &Rformat..; 
 class response / preloadfmt order=internal; 
 class item / order=&order; 
 classlev item / S=[just=l cellwidth=105 vjust=T]; 
 var response2; 
 classlev response/ style=<parent>; 
 table item=' '*[style=<parent>[just=c]],  response=' '*(pctn 
<response> ='%'*f=5.1)  
 response2=' '*(mean='Mean'*f=4.1 n='n'*f=4.)/ 
 RTS=50  printmiss misstext='0.0' BOX={LABEL="&statement" Style={JUST=l 
VJUST=t cellwidth=550}}; 
 title ' '; 
run; 
%end; 
 
%if &mean=0 %then %do; 
options orientation=landscape; 
 
proc tabulate data=longtable order=&order out=temptable; 
 format item &Iformat.. response &Rformat.. response2 &Rformat..; 
 class response / preloadfmt order=internal; 
 class item / order=&order; 
 classlev item / S=[just=l cellwidth=105 vjust=T]; 
 var response2; 
 classlev response/ style=<parent>; 
 table item=' '*[style=<parent>[just=c]],  response=' '*(pctn 
<response> ='%'*f=5.1)  
 response2=' '*n='n'*f=4./ 
 RTS=50  printmiss misstext='0.0' BOX={LABEL="&statement" Style={JUST=l 
VJUST=t cellwidth=550}}; 
 title ' '; 
run; 
%end; 
%end; 
 
%mend surveyreport; 
*End of surveyreport macro; 
 
*To call surveyreport macro - 
%surveyreport(dataset,scale,mean,combine,first,last,Rformat,Iformat,category
,statement,order); 

/************************************************************************** 
/*  STEP 4:  UPDATE AND RUN THE LONGREPORT MACRO CODE FOR DESIRED TABLES 
*************************************************************************** 
*/’ 
 
*Start of longreport macro; 
%macro longreport(filename,path,fileformat); 
option nodate nonumber; 
ods &fileformat file="&path.&filename..&fileformat" style=OPE; 
 
*Add calls to surveyreport macro for as many subscale as are on your survey;
%surveyreport(beta,53,0,1,3,16,howoften,questiontwo,"Monthly + Weekly + 
Daily",Considering the current school year indicate how often your students 
do each of the following.,data) 
%surveyreport(beta,42,1,1,17,23,agreement,questionthree,"Agree + Strongly 
Agree",Thinking about this school year to what extent do you agree with the 
following statements? I intentionally...,data) 
 
ods &fileformat close; 
%mend longreport; 
*End of longreport macro; 
 
*To call longreport macro - 
%longreport(FILENAME, PATH TO SAVE FILE, csv); *Produces a spreadsheet; 

%longreport(FILENAME, PATH TO SAVE FILE, rtf); *Produces a word processing 
document; 

 

 


